

# Porting of drivers/fmc/fmc-fakedev.c

May 26, 2017

## 1 Introduction

The driver `drivers/fmc/fmc-fakedev.c` was introduced in 2013 in the commit `6c62a89`. It is a fake software carrier driver. This driver was accompanied by changes in documentation, `Kconfig` and a make file, as well as a new documentation file. `Gcc` produces 2 errors that reduce to the same 2 errors.

## 2 struct `fmc_operations.read32`

`Gcc` reports that the field `read32` of the structure `struct fmc_operations` is not known. We try the following patch query (`step1.cocci`):

```
@bad1 depends on before@
struct fmc_operations *e;
@@

    e->read32

@bad2 depends on before@
struct fmc_operations e;
@@

    e.read32

@bad3 depends on before@
identifier i;
expression e;
@@

struct fmc_operations i = {
    .read32
    = e,
};

@depends on !bad1 && !bad2 && !bad3@
struct fmc_operations *e;
@@

+ e->read32

@depends on !bad1 && !bad2 && !bad3@
struct fmc_operations e;
@@
```

```

+ e.read32

@depends on !bad1 && !bad2 && !bad3@
identifier i;
expression e;
@@

struct fmc_operations i = {
+ .read32
  = e,
};

```

The only commit is c2955da at 18%. The commit message and code clearly show that the `read32` field should be backported to `readl`.

The following is more general than needed, because the driver only contains the structure initialization case, but in terms of prequel, the other cases also show what to do.

```

@@
struct fmc_operations *e;
@@

- e->read32
+ e->readl

@@
struct fmc_operations e;
@@

- e.read32
+ e.readl

@@
identifier i;
expression e;
@@

struct fmc_operations i = {
- .read32
+ .readl
  = e,
};

```

=== success 1/1, c2955da

### 3 struct fmc\_operations.write32

Gcc reports that the field `write32` of the structure `struct fmc_operations` is not known. We try the following patch query (step2.cocci):

```

@bad1 depends on before@
struct fmc_operations *e;
@@

```

```

e->write32

@bad2 depends on before@
struct fmc_operations e;
@@

e.write32

@bad3 depends on before@
identifier i;
expression e;
@@

struct fmc_operations i = {
    .write32
    = e,
};

@depends on !bad1 && !bad2 && !bad3@
struct fmc_operations *e;
@@

+ e->write32

@depends on !bad1 && !bad2 && !bad3@
struct fmc_operations e;
@@

+ e.write32

@depends on !bad1 && !bad2 && !bad3@
identifier i;
expression e;
@@

struct fmc_operations i = {
+ .write32
    = e,
};

```

The only commit is the same as in the previous case. This time, `--tt write32` should be backported to `writel`,

```

@@
struct fmc_operations *e;
@@

- e->write32
+ e->writel

@@
struct fmc_operations e;
@@

- e.write32
+ e.writel

```

```
@@
identifier i;
expression e;
@@

struct fmc_operations i = {
- .write32
+ .writel
  = e,
};
```

=== success 1/1, c2955da